
ΚΕΦΑΛΑΙΟ 12

Τεχνικές προσομοίωσης

Εισαγωγή

Σε πολλά πρακτικά προβλήματα η συμπεριφορά του εξεταζόμενου συστήματος είναι τόσο περίπλοκη, ώστε να είναι αδύνατη η ανάλυσή του με αναλυτικές μεθόδους. Η πολυπλοκότητα του συστήματος συνήθως οδηγεί στη δυσκολία διατύπωσης και επίλυσης των σχέσεων που διέπουν τη δυναμική συμπεριφορά του (μεταβολές της κατάστασής του) κάτω από την επίδραση διαφόρων παραγόντων. Για την αντιμετώπιση τέτοιων προβλημάτων, συχνά χρησιμοποιείται η τεχνική της **προσομοίωσης**.

Η βασική ιδέα της προσομοίωσης είναι η κατασκευή ενός μοντέλου το οποίο να μιμείται το πραγματικό σύστημα στις συνθήκες που ενδιαφέρουν. Με τη βοήθεια του μοντέλου μπορούν να εφαρμοστούν πειραματικές πολιτικές, προκειμένου αυτές να αξιολογηθούν (με διάφορα κριτήρια) και να επιλεγεί η καταλληλότερη.

Η προσομοίωση δεν περιορίζεται σε προβλήματα υπό συνθήκες βεβαιότητας. Αντίθετα, οι τεχνικές προσομοίωσης εφαρμόζονται κυρίως για την ανάλυση προβλημάτων με αβεβαιότητα. Σε πολλά προβλήματα του τύπου αυτού το πλήθος των αβέβαιων μεταβλητών είναι μεγάλο για να επιλυθούν με αναλυτικές τεχνικές.

Το πρώτο βήμα για την εφαρμογή της προσομοίωσης σε συνθήκες αβεβαιότητας είναι η ανάπτυξη ενός κατάλληλου μοντέλου, το οποίο να περιγράφει με όσο το δυνατό μεγαλύτερη ακρίβεια το φυσικό πρόβλημα. Κατά την ανάπτυξη του μοντέλου οι αβέβαιες μεταβλητές εισόδου αντικαθίστανται με τυχαίες τιμές (παρατηρήσεις). Με τη βοήθεια του μοντέλου είναι δυνατό να εκτελεστεί μια σειρά πειραμάτων, μεταβάλλοντας τις τυχαίες παρατηρήσεις μέσα στο εύρος των επιτρεπτών ορίων. Βέβαια, οι παρατηρήσεις που δη-

μιουργούνται με τη βοήθεια των τυχαίων τιμών πρέπει να συμπεριφέρονται στατιστικά όπως και οι πραγματικές παρατηρήσεις (οι παρατηρήσεις που θα προέκυπταν αν το γεγονός συνέβαινε πραγματικά). Αυτό σημαίνει ότι η ακολουθία των τυχαίων παρατηρήσεων κάθε μεταβλητής πρέπει να είναι σύμφωνη με την κατανομή πιθανότητας των πραγματικών παρατηρήσεων. Η προσομοίωση που στηρίζεται σε κατανομές πιθανότητας και τυχαίους αριθμούς ονομάζεται προσομοίωση **Monte Carlo** (από το γνωστό καζίνο του Monaco).

Αν και η προσομοίωση Monte Carlo δεν είναι μία τεχνική αριστοποίησης, τα πειράματα με τυχαίες παρατηρήσεις επιτρέπουν στους αποφασίζοντες να κατανοήσουν το φυσικό πρόβλημα και να αναδείξουν τις περισσότερο ή λιγότερο πιθανές καταστάσεις. Με στατιστική επεξεργασία των αποτελεσμάτων των πειραμάτων είναι δυνατό να εκτιμηθούν οι αναμενόμενες τιμές, οι τυπικές αποκλίσεις και τα διαστήματα εμπιστοσύνης μεγεθών που αποτελούν κριτήρια αξιολόγησης. Επομένως, η προσομοίωση δεν οδηγεί στον εντοπισμό της άριστης λύσης αλλά επιτρέπει τη ανάλυση εναλλακτικών σεναρίων. Δηλαδή η προσομοίωση είναι κατάλληλη για “what-if” αναλύσεις πολιτικών, όπου αλλάζοντας τις παραμέτρους του συστήματος προκύπτουν οι επιπτώσεις της αλλαγής αυτής στα κριτήρια αξιολόγησης.

Στο κεφάλαιο αυτό παρουσιάζεται ο τρόπος εφαρμογής τεχνικών προσομοίωσης Monte Carlo με τη βοήθεια του Excel, με την επίλυση δύο πρακτικών προβλημάτων.

Τυχαίοι αριθμοί

Για την παραγωγή τυχαίων αριθμών το Excel διαθέτει τη συνάρτηση **RAND**. Η συνάρτηση αυτή δεν παίρνει κανένα όρισμα και επιστρέφει ένα τυχαίο αριθμό μεταξύ 0 και 1 (συμπεριλαμβανομένων των 0 και 1). Οι τιμές που επιστρέφει η **RAND** έχουν δύο ιδιότητες:

1. Είναι ομοιόμορφα κατανεμημένες.
2. Είναι στατιστικά ανεξάρτητες.

Η πρώτη ιδιότητα σημαίνει ότι κάθε τιμή μεταξύ 0 και 1 έχει την ίδια πιθανότητα να εμφανιστεί. Με άλλα λόγια, δεν υπάρχουν ευνοούμενοι αριθμοί, αριθμοί δηλαδή που να εμφανίζονται πιο συχνά από όσο αναμένεται με την κανονική εναλλαγή της τύχης. Η δεύτερη ιδιότητα σημαίνει ότι η εμφάνιση ενός αριθμού δεν εξαρτάται από τους αριθμούς που έχουν προκύψει κατά τις προηγούμενες χρήσεις της συνάρτησης. Με άλλα λόγια, κάθε φορά που χρησιμοποιείται η συνάρτηση **RAND** υπάρχει η ίδια πιθανότητα να εμφανιστεί ένας συγκεκριμένος αριθμός.

Αν και με τη πρώτη ματιά, οι περιπτώσεις εφαρμογής της συνάρτησης RAND φαίνονται πολύ περιορισμένες (δηλαδή μόνο σε τυχαίες μεταβλητές που ακολουθούν ομοιόμορφη κατανομή στο διάστημα από 0 έως 1), αυτό δε συμβαίνει στην πράξη. Με κατάλληλους χειρισμούς οι τιμές της συνάρτησης μπορούν να χρησιμοποιηθούν για την παραγωγή τυχαίων αριθμών που να ακολουθούν οποιαδήποτε κατανομή. Στα παραδείγματα που ακολουθούν εξετάζονται τέτοιες περιπτώσεις.

Πρόβλημα παραγγελιών

Στην παράγραφο αυτή θα εξεταστεί ένα απλό παράδειγμα προσομοίωσης ενός προβλήματος με μια διακριτή τυχαία μεταβλητή. Στόχος είναι η αποσαφήνιση των εννοιών που αφορούν στην προσομοίωση και η εκμάθηση του τρόπου εφαρμογής της στο Excel.

Κατά το μήνα Αύγουστο, ο ιδιοκτήτης ενός βιβλιοπωλείου πρέπει να αποφασίσει πόσα ημερολόγια του ερχόμενου έτους θα παραγγείλει. Κάθε ημερολόγιο κοστίζει 7 € και πωλείται προς 9,5 €. Μετά το τέλος του Ιανουαρίου όλα τα αδιάθετα ημερολόγια επιστρέφονται στον εκδότη προς 2,4 € το καθένα. Ο βιβλιοπώλης έχει εκτιμήσει τη ζήτηση των ημερολογίων στα επίπεδα που φαίνεται στον πίνακα 12.1. Στόχος του βιβλιοπώλη είναι να μεγιστοποιήσει το κέρδος παραγγέλλοντας τον κατάλληλο αριθμό ημερολογίων

Πίνακας 12.1 Κατανομή της ζήτησης των ημερολογίων.

Ημερολόγια	Πιθανότητα
100	0,30
150	0,20
200	0,30
250	0,15
300	0,05

Λύση

Το πρόβλημα είναι αρκετά απλό καθώς η μόνη αβέβαιη μεταβλητή είναι η ζήτηση των ημερολογίων. Επιλέγοντας μια συγκεκριμένη ποσότητα παραγγελίας, μπορεί να σχηματιστεί ένας αριθμός (έστω 50) τυχαίων τιμών ζήτησης και να υπολογιστεί η μέση (αναμενόμενη) τιμή των κερδών.

Για τον υπολογισμό των τυχαιών τιμών της ζήτησης χρησιμοποιείται η συνάρτηση RAND. Προκειμένου οι τιμές που επιστρέφει η συνάρτηση αυτή να μετασχηματιστούν σε τυχαιές τιμές της ζήτησης, σύμφωνες με τη διακριτή κατανομή πιθανότητας του πίνακα 12.1, ακολουθείται η παρακάτω τακτική. Το διάστημα $[0, 1]$ διαιρείται σε πέντε υποδιαστήματα, με μήκη ίσα με τις πιθανότητες των αντίστοιχων δυνατών τιμών της ζήτησης (βλ. σχήμα 12.1). Για κάθε τυχαία τιμή που επιστρέφει η RAND, βρίσκεται το υποδιάστημα στο οποίο αυτή ανήκει και επιλέγεται η τιμή της ζήτησης στην οποία αντιστοιχεί το συγκεκριμένο υποδιάστημα. Για παράδειγμα, αν η τυχαία τιμή είναι μεταξύ 0 και 0,3 (μήκος 0,3) επιλέγεται η τιμή 100, αν είναι μεταξύ 0,3 και 0,5 (μήκος 0,2) επιλέγεται η τιμή 150, κλπ. Η διαδικασία αυτή επιτυγχάνεται στο Excel με τη βοήθεια της συνάρτησης VLOOKUP και ενός πίνακα αντιστοιχίας, όπως εξηγείται στη συνέχεια.

	0	0,3	0,5	0,8	0,95	1
μήκος	0,3	0,2	0,3	0,15	0,05	
ζήτηση	100	150	200	250	300	

Σχήμα 12.1 Παραγωγή τυχαιών τιμών της ζήτησης.

Το μοντέλο βρίσκεται στο φύλλο εργασίας Model του αρχείου BOOKSTORE.XLS, μέρος του οποίου παρουσιάζεται στο σχήμα 12.2.

- ▶ **Δεδομένα εισόδου.** Στην περιοχή B4:B6 εισάγονται τα δεδομένα κόστους και στις περιοχές D5:D9 και F5:F9 η κατανομή της ζήτησης. Τέλος, στο κελί B9 εισάγεται η εξεταζόμενη ποσότητα παραγγελίας.
- ▶ **Αθροιστικές πιθανότητες ζήτησης.** Στη στήλη E υπολογίζονται οι αθροιστικές πιθανότητες, γεγονός που θα επιτρέψει την παραγωγή των τυχαιών τιμών της ζήτησης. Στο κελί E5 εισάγεται ο αριθμός 0, στο κελί E6 ο τύπος:

```
=E5+D5
```

και αντιγράφεται προς τα κάτω μέχρι το κελί E9.

- ▶ **Παραγωγή τυχαιών αριθμών.** Οι τυχαιοί αριθμοί παράγονται στην περιοχή B13:B62. Έτσι στο κελί B13 εισάγεται ο τύπος:

```
=RAND()
```

και αντιγράφεται προς τα κάτω μέχρι το κελί B62.

	A	B	C	D	E	F	G
1	Προσομοίωση Βιβλιοπωλείου						
2							
3	Δεδομένα κόστους			Κατανομή ζήτησης			
4	Κόστος	7		Πιθανότητα	Αθροιστική	Ζήτηση	
5	Τιμή πώλησης	9,5		0,30	0,00	100	
6	Τιμή επιστροφής	2,4		0,20	0,30	150	
7				0,30	0,50	200	
8	Μεταβλητή απόφασης			0,15	0,80	250	
9	Ποσ. παραγγελίας	200		0,05	0,95	300	
10							
11	Προσομοίωση						
12	Επανάληψη	Τυχαίος αρ.	Ζήτηση	Έσοδα	Κόστη	Επιστροφές	Κέρδη
13	1	0,2510	100	950	1.400	240	-210
14	2	0,9434	250	1.900	1.400	0	500
15	3	0,7431	200	1.900	1.400	0	500
16	4	0,8130	250	1.900	1.400	0	500
17	5	0,9753	300	1.900	1.400	0	500
18	6	0,3057	150	1.425	1.400	120	145
19	7	0,7923	200	1.900	1.400	0	500
20	8	0,0811	100	950	1.400	240	-210
21	9	0,0680	100	950	1.400	240	-210
22	10	0,0739	100	950	1.400	240	-210
52	40	0,0521	100	950	1.400	240	-210
53	41	0,2405	100	950	1.400	240	-210
54	42	0,0475	100	950	1.400	240	-210
55	43	0,3272	150	1.425	1.400	120	145
56	44	0,8135	250	1.900	1.400	0	500
57	45	0,1253	100	950	1.400	240	-210
58	46	0,2566	100	950	1.400	240	-210
59	47	0,9227	250	1.900	1.400	0	500
60	48	0,1032	100	950	1.400	240	-210
61	49	0,2980	100	950	1.400	240	-210
62	50	0,2755	100	950	1.400	240	-210
63							
64	Στατιστικά μεγέθη						
65	Μέση τιμή κερδών	124		95% διάστημα εμπιστοσύνης για αναμενόμενα κέρδη			
66	Τυπ. απόκλ. κερδών	324		Πάνω όριο	32		
67	Ελάχιστο κέρδος	-210		Κάτω όριο	216		
68	Μέγιστο κέρδος	500					

Σχήμα 12.2 Προσομοίωση βιβλιοπωλείου.

- **Τυχαίες τιμές ζήτησης.** Στο κελί C13 εισάγεται ο τύπος:

```
=VLOOKUP(B13;$E$5:$F$9;2)
```

και αντιγράφεται προς τα κάτω μέχρι το κελί C62. Η συνάρτηση VLOOKUP ψάχνει στις τιμές της πρώτης στήλης του πίνακα E5:F9 (δεύτερο όρισμα), δηλαδή τη στήλη με τις αθροιστικές πιθανότητες, για τη μεγαλύτερη τιμή που είναι μικρότερη (όχι ίση) από τον τυχαίο αριθμό του κελιού B13 (πρώτο όρισμα). Στη συνέχεια επιστρέφει την τιμή που βρίσκεται στην ίδια γραμμή με τον αριθμό αυτό και στη στήλη 2 (τρίτο όρισμα). Με την πρακτική αυτή μπορούν να δημιουρ-

γηθούν τυχαίοι αριθμοί από οποιαδήποτε διακριτή κατανομή πιθανότητας (η περιγραφή της συνάρτησης VLOOKUP δίνεται παρακάτω).

- ▶ **Έσοδα.** Τα έσοδα καθορίζονται από το επίπεδο της ζήτησης. Ο αριθμός των ημερολογίων που θα πουληθούν καθορίζεται από τη μικρότερη τιμή της ζήτησης και της ποσότητας παραγγελίας. Έτσι στο κελί D13 εισάγεται ο τύπος:

```
=B$5*MIN(C13;B$9)
```

και αντιγράφεται προς τα κάτω μέχρι το κελί D62.

- ▶ **Κόστη.** Το κόστος παραγγελίας είναι σταθερό και δεν εξαρτάται από τη ζήτηση. Στο κελί E13 εισάγεται ο τύπος:

```
=B$9*B$4
```

και αντιγράφεται προς τα κάτω μέχρι το κελί E62.

- ▶ **Επιστροφές.** Αν η ποσότητα παραγγελίας υπερβαίνει τη ζήτηση υπάρχει μία επιστροφή 2,4 € για κάθε αδιάθετο ημερολόγιο. Σε αντίθετη περίπτωση δεν υπάρχει επιστροφή. Η συνθήκη αυτή εκφράζεται με το ακόλουθο τύπο που εισάγεται στο κελί F13:

```
=B$6*MAX(B$9-C13;0)
```

ο οποίος αντιγράφεται προς τα κάτω μέχρι το κελί F62.

- ▶ **Κέρδη.** Το κέρδος υπολογίζεται εύκολα αφαιρώντας από τα έσοδα μείον το κόστος και προσθέτοντας τις επιστροφές. Στο κελί G13 εισάγεται ο τύπος:

```
=D13-E13+F13
```

και αντιγράφεται προς τα κάτω μέχρι το κελί G62.

- ▶ **Στατιστικά μεγέθη.** Στα κελιά B65:B68 υπολογίζονται η μέση τιμή, η τυπική απόκλιση, η ελάχιστη και η μέγιστη τιμή του κέρδους, ως:

```
=AVERAGE(G13:G62)
```

```
=STDEV(G13:G62)
```

```
=MIN(G13:G62)
```

```
=MAX(G13:G62)
```

- ▶ **Διάστημα εμπιστοσύνης.** Τέλος, στα κελιά E66 και E67 υπολογίζονται τα όρια του 95% διαστήματος εμπιστοσύνης για τη μέση τιμή του κέρδους, ως:

```
=B65-TINV(0,05;49)*B66/SQRT(50)
```

```
=B65+TINV(0,05;49)*B66/SQRT(50)
```

ΣΥΝΑΡΤΗΣΗ VLOOKUP

Η συνάρτηση VLOOKUP επιτρέπει την αναζήτηση πληροφοριών που βρίσκονται σε πίνακες. Λειτουργεί με κατακόρυφους πίνακες, δηλαδή πίνακες ταξινομημένους κατά στήλες.

Η σύνταξή της είναι η ακόλουθη:

VLOOKUP(value;table;column)

Το όρισμα όπου value αντιπροσωπεύει την τιμή που αναζητείται στην πρώτη στήλη του πίνακα, table είναι η περιοχή κελιών που περιέχει τον πίνακα και column είναι η στήλη του πίνακα από την οποία θα επιλεγεί το αποτέλεσμα.

Αν η VLOOKUP δεν εντοπίσει ακριβώς την τιμή value στον πίνακα table, χρησιμοποιεί τη μεγαλύτερη τιμή που είναι μικρότερη από την value.

Υπολογισμός της βέλτιστης ποσότητας παραγγελίας. Υπολογίζεται με τη βοήθεια του εργαλείου Πίνακας Δεδομένων, στην περιοχή A72:B81.

- ▶ Στο κελί B72 εισάγεται ο τύπος:


=B65

- ▶ Επιλέγεται η περιοχή A72:B81 και από το μενού Data ενεργοποιείται η επιλογή Table. Στο πλαίσιο διαλόγου που εμφανίζεται εισάγεται ως Column Input το κελί B9. Τα αποτελέσματα παρουσιάζονται στο σχήμα 12.3.

Οικονομική αξιολόγηση επένδυσης

Πολλές εταιρείες χρησιμοποιούν τεχνικές προσομοίωσης κατά την οικονομική ανάλυση επενδυτικών σχεδίων, προκειμένου να ληφθεί υπόψη η αβεβαιότητα για τις διάφορες χρηματοροές. Ακολουθεί ένα χαρακτηριστικό παράδειγμα.

Μια αυτοκινητοβιομηχανία σχεδιάζει την παραγωγή ενός νέου μοντέλου αυτοκινήτου, το οποίο πρόκειται να διαθέσει στην αγορά για τα επόμενα δέκα έτη. Οι υπεύθυνοι της βιομηχανίας, προκειμένου να αξιολογήσουν το σχέδιο παραγωγής του νέου μοντέλου, έχουν κάνει προβλέψεις για το σταθερό κόστος της επένδυσης, το μεταβλητό κόστος παραγωγής κάθε αυτοκινήτου, την τιμή πώλησης κάθε αυτοκινήτου και τη ζήτηση για τα επόμενα δέκα έτη.



Σχήμα 12.3 Υπολογισμός βέλτιστης ποσότητας παραγγελίας.

Το σταθερό κόστος επένδυσης δεν είναι γνωστό με βεβαιότητα. Έχει εκτιμηθεί ότι ακολουθεί κανονική κατανομή με μέση τιμή 1.500 εκατομμύρια € και τυπική απόκλιση 200 εκατομμύρια €. Όλο το σταθερό κόστος πραγματοποιείται στην αρχή του πρώτου έτους.

Το μεταβλητό κόστος στο πρώτο έτος προσεγγίζεται, επίσης, με μια κανονική κατανομή με μέση τιμή 5,5 χιλιάδες € ανά αυτοκίνητο και τυπική απόκλιση 0,18 χιλιάδες €. Το μεταβλητό κόστος για κάθε ένα από τα επόμενα έτη υπολογίζεται από το μεταβλητό κόστος του προηγούμενου έτους, αφού πολλαπλασιαστεί με ένα συντελεστή πληθωρισμού. Ο συντελεστής πληθωρισμού ακολουθεί κανονική κατανομή με μέση τιμή 1,05 και τυπική απόκλιση 0,02. Όλα τα μεταβλητά κόστη πραγματοποιούνται στο τέλος του αντίστοιχου έτους.

Η τιμή πώλησης για το πρώτο έτος καθορίζεται στις 8,3 χιλιάδες € ανά αυτοκίνητο. Για τα επόμενα έτη, η τιμή πώλησης πολλαπλασιάζεται με το συντελεστή πληθωρισμού που περιγράφηκε παραπάνω. Όπως τα μεταβλητά κόστη, τα έσοδα από τις πωλήσεις πραγματοποιούνται στο τέλος του αντίστοιχου έτους.

Η ζήτηση των αυτοκινήτων για το πρώτο έτος προσεγγίζεται από μια κανονική κατανομή με μέση τιμή 100.000 και τυπική απόκλιση 10.000 αυτοκίνητα. Για τα επόμενα έτη η ζήτηση ακολουθεί κανονική κατανομή με μέση τιμή την πραγματική ζήτηση του προηγούμενου έτους και τυπική απόκλιση 10.000.

Η πολιτική της βιομηχανίας είναι να στηρίζει τον όγκο παραγωγής στην κατανομή πιθανότητας της ζήτησης κάθε έτους. Συγκεκριμένα, αν η αναμε-

	A	B	C	D	E	F	G	H	I	J	K	L
1	Προσομοίωση Παραγωγής Νέου Αυτοκινήτου											
2												
3	Σταθερό κόστος (εκ. Ευρο)						Ζήτηση 1ου έτους (χιλ. αυτοκίνητα)					
4	Κανονική Κατανομή						Κανονική Κατανομή					
5	Μέση Τιμή	1.500					Μέση Τιμή	100				
6	Τυπ. Απ.	200					Τυπ. Απ.	10				
7												
8	Μεταβλητό κόστος παραγωγής 1ου έτους (χιλ. Ευρο ανά αυτοκίνητο)						Ζήτηση ετών 2-10 (χιλ. αυτοκίνητα)					
9	Κανονική Κατανομή						Κανονική Κατανομή					
10	Μέση Τιμή	5,5					Μέση Τιμή	Ζήτηση προηγούμενου έτους				
11	Τυπ. Απ.	0,18					Τυπ. Απ.	10				
12												
13	Μεταβλητό κόστος παραγωγής ετών 2-10											
14	Όσο με το μεταβλητό κόστος του προηγούμενου έτους											
15	πολλαπλασιασμένο με τον συντελεστή πληθωρισμού.											
16												
17	Τιμή πώλησης 1ου έτους (χιλ. Ευρο ανά αυτοκίνητο)						Πολιτική παραγωγής					
18	Σταθερή	8,3					Μεταβλ. κ	1				
19							Εκπτώση %					
20	Τιμή πώλησης ετών 2-10						30%					
21	Όση με την τιμή πώλησης του προηγούμενου έτους						Επιτόκιο προεξόφλησης					
22	πολλαπλασιασμένη με τον συντελεστή πληθωρισμού.						9%					
23												
24	Συντελεστής πληθωρισμού ετών 2-10											
25	Κανονική Κατανομή											
26	Μέση Τιμή	1,05										
27	Τυπ. Απ.	0,02										
28												
29	Μοντέλο											
30	Έτος		1	2	3	4	5	6	7	8	9	10
31	Συντελ. πληθωρισμού			1,06	1,05	1,05	1,07	1,08	1,04	1,04	1,06	1,07
32	Ζήτηση (1000)		79,31	84,19	76,82	77,12	84,62	72,80	74,63	49,57	40,74	45,96
33	Παραγωγή (1000)		110,00	89,31	94,19	86,82	87,12	94,62	82,80	84,63	59,57	50,74
34	Μετ. Κόστος (1000 Ευρο)		5,66	5,99	6,29	6,61	7,05	7,62	7,92	8,22	8,72	9,34
35	Τιμή Πώλησης (1000 Ευρο)		8,30	8,78	9,22	9,69	10,33	11,18	11,61	12,05	12,78	13,70
36	Κόστος Παρ. (εκ. Ευρο)		622,57	534,97	591,97	573,69	613,96	721,12	655,55	695,50	519,34	474,00
37	Έσοδα (εκ. Ευρο)		836,59	771,03	820,11	813,13	892,63	984,37	932,89	893,22	689,39	675,50
38												
39	Σταθερό κόστος (εκ. Ευρο)		1.680,73									
40												
41	NPV (εκ. Ευρο)											
42	Κόστος παραγωγής		3.860,98									
43	Έσοδα		5.354,24									
44	Σύνολο		-187,46									

Σχήμα 12.4 Μοντέλο προσομοίωσης της επένδυσης της αυτοκινητοβιομηχανίας.

νόμηση ζήτηση για το έτος t είναι ίση με ED_t και η τυπική απόκλιση της ζήτησης είναι σ_t , τότε η βιομηχανία παράγει $ED_t + k \sigma_t$ αυτοκίνητα. Όπου k είναι μια μεταβλητή απόφασης. Αν η ζήτηση ενός έτους υπερβαίνει την παραγωγή, η επιπλέον ζήτηση χάνεται. Όμως, αν η παραγωγή είναι μεγαλύτερη από τη ζήτηση, τα επιπλέον αυτοκίνητα παλούνται στο τέλος του έτους με 30% έκπτωση.

Αν το επιτόκιο προεξόφλησης είναι 9% να αναπτυχθεί ένα μοντέλο που να υπολογίζει την καθαρή παρούσα αξία της επένδυσης.

Λύση

Το μοντέλο βρίσκεται στο αρχείο CAR.XLS (σχήμα 12.4). Αποτελείται από τα ακόλουθα στοιχεία:

- **Δεδομένα εισόδου.** Εισάγονται στην αρχή του φύλλου εργασίας, μέχρι τη γραμμή 27. Στο κελί H18 εισάγεται η τιμή 1 για τη μεταβλητή

απόφασης κ. Αργότερα η τιμή αυτή μπορεί να τροποποιηθεί προκειμένου να ελεγχθεί η επίδρασή της στα αποτελέσματα.

- ▶ **Συντελεστές πληθωρισμού.** Εισάγονται στη γραμμή 31. Ο συντελεστής πληθωρισμού για το πρώτο έτος δεν απαιτείται. Για τα επόμενα έτη ο συντελεστής πληθωρισμού υπολογίζεται με βάση την κανονική κατανομή. Έτσι στο κελί D31 εισάγεται ο τύπος:

```
=NORMINV(RAND();$B$26;$B$27)
```

και αντιγράφεται προς τα δεξιά μέχρι το κελί L31. Η παραπάνω σχέση αντιπροσωπεύει τον τρόπο υπολογισμού τυχαίων αριθμών που ακολουθούν κανονική κατανομή.

- ▶ **Ζήτηση αυτοκινήτων.** Η ζήτηση για το πρώτο έτος υπολογίζεται στο κελί C32 με τον τύπο:

```
=NORMINV(RAND();H5;H6)
```

Η ζήτηση για τα επόμενα έτη ακολουθεί κανονική κατανομή με μέση τιμή την πραγματική ζήτηση του προηγούμενου έτους. Έτσι στο κελί D32 εισάγεται ο τύπος:

```
=NORMINV(RAND();C32;$H$11)
```

και αντιγράφεται προς τα δεξιά μέχρι το κελί L32.

- ▶ **Όγκος παραγωγής.** Η παραγωγή για το πρώτο έτος υπολογίζεται στο κελί C33 από την αναμενόμενη ζήτηση του πρώτου έτους και την τυπική της απόκλιση:

```
=H5+H18*H6
```

Αντίστοιχα, η παραγωγή των επόμενων ετών στηρίζεται στην αναμενόμενη τιμή της ζήτησης, που είναι ίση με την πραγματική τιμή της ζήτησης κατά το προηγούμενο έτος. Έτσι στο κελί D33 εισάγεται ο τύπος:

```
=C32+$H$18*$H$11
```

και αντιγράφεται προς τα δεξιά μέχρι το κελί L33.

- ▶ **Μεταβλητό κόστος παραγωγής.** Για το πρώτο έτος το μεταβλητό κόστος παραγωγής υπολογίζεται, στο κελί C34, με βάση την κανονική κατανομή ως:

```
=NORMINV(RAND();B10;B11)
```

Για τα επόμενα έτη το κόστος παραγωγής πολλαπλασιάζεται με το συντελεστή πληθωρισμού. Έτσι στο κελί D34 εισάγεται ο τύπος:

=C34*D31

και αντιγράφεται προς τα δεξιά μέχρι το κελί L34.

- ▶ **Τιμή πώληση.** Είναι σταθερή για το πρώτο έτος. Στο κελί C35 εισάγεται ο τύπος:

=B18

Για τα επόμενα έτη η τιμή πώλησης πολλαπλασιάζεται με το συντελεστή πληθωρισμού. Έτσι στο κελί D35 εισάγεται ο τύπος:

=C35*D31

και αντιγράφεται προς τα δεξιά μέχρι το κελί L35.

- ▶ **Συνολικό κόστος παραγωγής.** Είναι ίσο με τον όγκο παραγωγής πολλαπλασιασμένο με το μοναδιαίο μεταβλητό κόστος. Στο κελί C36 εισάγεται ο τύπος:

=C33*C34

και αντιγράφεται προς τα δεξιά μέχρι το κελί L36.

- ▶ **Έσοδα.** Αν η ζήτηση υπερβαίνει τον όγκο παραγωγής τότε τα έσοδα είναι ίσα με την τιμή πώλησης επί τον όγκο παραγωγής. Διαφορετικά, αν ο όγκος παραγωγής υπερβαίνει τη ζήτηση, τότε τα έσοδα είναι ίσα με την τιμή πώλησης επί τη ζήτηση, συν την τιμή έκπτωσης επί τον αριθμό των επιπλέον αυτοκινήτων. Έτσι στο κελί C37 εισάγεται ο τύπος:

=IF(C33<C32;C35*C33;C35*(C32+(1-\$H\$19)*(C33-C32)))

και αντιγράφεται προς τα δεξιά μέχρι το κελί L37.

- ▶ **Σταθερό κόστος.** Υπολογίζεται στο κελί C39 (σε εκατομμύρια €) με βάση την κανονική κατανομή:

=NORMINV(RAND();B5;B6)

- ▶ **Παρούσες αξίες.** Οι παρούσες αξίες του μεταβλητού κόστους παραγωγής και των εσόδων υπολογίζονται στα κελιά C42 και C43 ως:

=NPV(H22;C36:L36)

=NPV(H22;C37:L37)

- ▶ **Καθαρή παρούσα αξία επένδυσης.** Υπολογίζεται στο κελί C44 ως:

=C43-C42-C39

Σημειώνεται ότι το σταθερό κόστος επένδυσης δεν προεξοφλείται καθώς πραγματοποιείται στην αρχή του έτους 1.

	A	B	C	D	E	F
1	Προσομοίωση Παραγωγής Νέου Αυτοκινήτου					
45						
46	Προσομοίωση					
47	Επανάληψη	Συνολ. NPV				
48		-187,46		Μέση Τιμή	697,51	
49	1	379,23		Τυπ. Απ.	471,31	
50	2	231,22		Ελάχιστη	-1.103,02	
51	3	1.037,84		Μέγιστη	2.673,36	
52	4	1.480,04				
53	5	705,70				
54	6	414,85				
55	7	47,31				
56	8	930,28				
57	9	1.108,05				
58	10	81,54				
59	11	2.278,56				
60	12	548,32				
61	13	-193,18				
62	14	760,34				
63	15	1.362,35				
64	16	197,70				
65	17	694,25				
66	18	1.478,05				
67	19	858,54				
68	20	537,24				
69	21	75,66				
70	22	318,34				
71	23	882,94				
72	24	605,56				
73	25	560,69				
74	26	456,47				
75	27	628,54				
76	28	1.351,59				
77	29	193,43				
78	30	380,20				

Σχήμα 12.5 Προσομοίωση παραγωγής νέου αυτοκινήτου.

Η προσομοίωση πραγματοποιείται στην περιοχή A48:B148 με τη βοήθεια του εργαλείου Πίνακας Δεδομένων.

- ▶ Στο κελί B48 εισάγεται ο τύπος:

=C44

που αναφέρεται στην καθαρή παρούσα αξία της επένδυσης.

- ▶ Επιλέγεται η περιοχή A48:B148 και ενεργοποιείται η επιλογή Data/Table. Στο πλαίσιο διαλόγου που εμφανίζεται εισάγεται ως Column Input ένα οποιοδήποτε κενό κελί.
- ▶ **Στατιστικά μεγέθη.** Στα κελιά E48:E51 υπολογίζονται η μέση τιμή, η τυπική απόκλιση, η ελάχιστη και η μέγιστη τιμή της καθαρής παρούσας αξίας, με βάση τις τιμές της προσομοίωσης:

```
=AVERAGE(B49:B148)  
=STDEV(B49:B148)  
=MIN(B49:B148)  
=MAX(B49:B148)
```

Τα αποτελέσματα παρουσιάζονται στο σχήμα 12.5, απ' όπου φαίνεται ότι η μέση τιμή της καθαρής παρούσας αξίας είναι ίση με 697,51 εκατομμύρια €. Τονίζεται ότι η τιμή αυτή είναι μια προσέγγιση της πραγματικής μέσης τιμής. Η ακρίβεια της προσέγγισης είναι μεγαλύτερη όσο μεγαλύτερος είναι ο αριθμός των τυχαίων πειραμάτων της προσομοίωσης.

Βιβλιογραφία

- Albright, S.C., Winston, W.L., and Zappe, C. (1999) *Data Analysis & Decision Making with Microsoft Excel*, Duxbury Press, USA.
- Barlow, F.G. (1999) *Excel Models for Business and Operations Management*, John Wiley & Sons, Chichester, Sussex.
- Gottfried, B.S (1998) *Spreadsheet Tools for Engineers*, McGraw-Hill, Singapore.
- Monks, J.G. (1987) *Operations Management*, Mc-Graw Hill, USA.
- Πραστάκος, Γ. (2000) *Διοικητική Επιστήμη: Λήψη Επιχειρηματικών Αποφάσεων στην Κοινωνία της Πληροφορίας*, Εκδόσεις Αθαν. Σταμούλης, Αθήνα.